NAG Library Routine Document

G01HDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01HDF returns a probability associated with a multivariate Student's t-distribution.

2 Specification

```
FUNCTION G01HDF (N, TAIL, A, B, NU, DELTA, ISCOV, RC, LDRC, EPSABS, EPSREL,     &
                 NUMSUB, NSAMPL, FMAX, ERREST, IFAIL)
REAL (KIND=nag_wp) G01HDF
INTEGER            N, ISCOV, LDRC, NUMSUB, NSAMPL, FMAX, IFAIL
REAL (KIND=nag_wp) A(N), B(N), NU, DELTA(N), RC(LDRC,N), EPSABS, EPSREL,       &
                   ERREST
CHARACTER(1)       TAIL(N)
```

3 Description

A random vector $x \in \mathbb{R}^n$ that follows a Student's t-distribution with ν degrees of freedom and covariance matrix Σ has density:

$$\frac{\Gamma((\nu+n)/2)}{\Gamma(\nu/2)\nu^{n/2}\pi^{n/2}|\Sigma|^{1/2}\left[1+\frac{1}{\nu}x^{\mathrm{T}}\Sigma^{-1}x\right]^{(\nu+n)/2}},$$

and probability p given by:

$$p=\frac{\Gamma((\nu+n)/2)}{\Gamma(\nu/2)\sqrt{|\Sigma|(\pi\nu)^n}}\int_{a_1}^{b_1}\int_{a_2}^{b_2}\cdots\int_{a_n}^{b_n}\left(1+x^{\mathrm{T}}\Sigma^{-1}x/\nu\right)^{-(\nu+n)/2}dx.$$

The method of calculation depends on the dimension n and degrees of freedom ν. The method of Dunnet and Sobel is used in the bivariate case if ν is a whole number. A Plackett transform followed by quadrature method is adopted in other bivariate cases and trivariate cases. In dimensions higher than three a number theoretic approach to evaluating multidimensional integrals is adopted.

Error estimates are supplied as the published accuracy in the Dunnet and Sobel case, a Monte–Carlo standard error for multidimensional intergrals, and otherwise the quadrature error estimate.

A parameter δ allows for non-central probabilities. The number theoretic method is used if any δ is nonzero.

In cases other than the central bivariate with whole ν, G01HDF attempts to evaluate probabilities within a requested accuracy $\max(\epsilon_a, \epsilon_r \times I)$, for an approximate integral value I, absolute accuracy ϵ_a and relative accuracy ϵ_r.

4 References

Dunnet C W and Sobel M (1954) A bivariate generalization of Student's t-distribution, with tables for certain special cases *Biometrika* **41** 153–169

Genz A and Bretz F (2002) Methods for the computation of multivariate t-probabilities *Journal of Computational and Graphical Statistics* **(11)** 950–971

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of dimensions.

Constraint: $1 < \mathrm{N} < 1000$.

2: TAIL(N) – CHARACTER(1) array *Input*

On entry: defines the calculated probability, set TAIL(i) to:

TAIL(i) = 'L'
If the ith lower limit a_i is negative infinity.

TAIL(i) = 'U'
If the ith upper limit b_i is infinity.

TAIL(i) = 'C'
If both a_i and b_i are finite.

Constraint: TAIL(i) = 'L', 'U' or 'C', for $i = 1, 2, \ldots, \mathrm{N}$.

3: A(N) – REAL (KIND=nag_wp) array *Input*

On entry: a_i, for $i = 1, 2, \ldots, n$, the lower integral limits of the calculation.

If TAIL(i) = 'L', A(i) is not referenced and the ith lower limit of integration is $-\infty$.

4: B(N) – REAL (KIND=nag_wp) array *Input*

On entry: b_i, for $i = 1, 2, \ldots, n$, the upper integral limits of the calculation.

If TAIL(i) = 'U', B(i) is not referenced and the ith upper limit of integration is ∞.

Constraint: if TAIL(i) = 'C', B$(i) >$ A(i).

5: NU – REAL (KIND=nag_wp) *Input*

On entry: ν, the degrees of freedom.

Constraint: $\mathrm{NU} > 0.0$.

6: DELTA(N) – REAL (KIND=nag_wp) array *Input*

On entry: DELTA(i) the noncentrality parameter for the ith dimension, for $i = 1, 2, \ldots, \mathrm{N}$; set DELTA$(i) = 0$ for the central probability.

7: ISCOV – INTEGER *Input*

On entry: set $\mathrm{ISCOV} = 1$ if the covariance matrix is supplied and $\mathrm{ISCOV} = 2$ if the correlation matrix is supplied.

Constraint: $\mathrm{ISCOV} = 1$ or 2.

8: RC(LDRC,N) – REAL (KIND=nag_wp) array *Input/Output*

On entry: the lower triangle of the matrix must contain the covariance matrix if $\mathrm{ISCOV} = 1$ or the correlation matrix if $\mathrm{ISCOV} = 2$.

On exit: the strict upper triangle of RC contains the correlation matrix used in the calculations.

9: LDRC – INTEGER *Input*

On entry: the first dimension of the array RC as declared in the (sub)program from which G01HDF is called.

Constraint: $\mathrm{LDRC} \geq \mathrm{N}$.

10: EPSABS – REAL (KIND=nag_wp) *Input*

On entry: ϵ_a, the absolute accuracy requested in the approximation. If EPSABS is negative, the absolute value is used.

Suggested value: 0.0.

11: EPSREL – REAL (KIND=nag_wp) *Input*

On entry: ϵ_r, the relative accuracy requested in the approximation. If EPSREL is negative, the absolute value is used.

Suggested value: 0.001.

12: NUMSUB – INTEGER *Input*

On entry: if quadrature is used, the number of sub-intervals; otherwise NUMSUB is not referenced.

Suggested value: 350.

Constraint: if referenced, $\text{NUMSUB} > 0$.

13: NSAMPL – INTEGER *Input*

On entry: if quadrature is used, NSAMPL is not referenced; otherwise NSAMPL is the number of samples used to estimate the error in the approximation.

Suggested value: 8

Constraint: if referenced, $\text{NSAMPL} > 0$.

14: FMAX – INTEGER *Input*

On entry: if a number theoretic approach is used, the maximum number of evaluations for each integrand function.

Suggested value: $1000 \times \text{N}$

Constraint: if referenced, $\text{FMAX} \geq 1$.

15: ERREST – REAL (KIND=nag_wp) *Output*

On exit: an estimate of the error in the calculated probability.

16: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\text{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, N $= \langle value \rangle$.
Constraint: N ≥ 2 and N ≤ 1000.

IFAIL $= 2$

On entry, TAIL$(k) = \langle value \rangle$.
Constraint: TAIL$(k) =$ 'L', 'U' or 'C'.

IFAIL $= 4$

On entry, $k = \langle value \rangle$.
Constraint: B$(k) \leq$ A(k) for a central probability.

IFAIL $= 5$

On entry, NU $= \langle value \rangle$.
Constraint: degrees of freedom NU > 0.0.

IFAIL $= 8$

On entry, ISCOV $= \langle value \rangle$.
Constraint: ISCOV $= 1$ or 2.

IFAIL $= 9$

On entry, the information supplied in RC is invalid.

IFAIL $= 10$

On entry, LDRC $= \langle value \rangle$ and N $= \langle value \rangle$.
Constraint: LDRC $\geq$ N.

IFAIL $= 12$

On entry, NUMSUB $= \langle value \rangle$.
Constraint: NUMSUB ≥ 1.

IFAIL $= 13$

On entry, NSAMPL $= \langle value \rangle$.
Constraint: NSAMPL ≥ 1.

IFAIL $= 14$

On entry, FMAX $= \langle value \rangle$.
Constraint: FMAX ≥ 1.

7 Accuracy

An estimate of the error in the calculation is given by the value of ERREST on exit.

8 Further Comments

None.

9 Example

This example prints two probabilities from the Student's t-distribution.

9.1 Program Text

```
    Program g01hdfe

!      G01HDF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: g01hdf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                 :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)                 :: epsabs, epsrel, errest, nu, prob
      Integer                            :: fmax, i, ifail, iscov, ldrc, n,      &
                                            nsampl, numsub
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable    :: a(:), b(:), delta(:), rc(:,:)
      Character (1), Allocatable         :: tail(:)
!     .. Executable Statements ..
      Write (nout,*) 'G01HDF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n, iscov

      ldrc = n

      numsub = 200
      nsampl = 8
      fmax = 25000
      epsabs = 0.0E0_nag_wp
      epsrel = 1.0E-3_nag_wp

      Allocate (tail(n),a(n),b(n),delta(n),rc(ldrc,n))

d_lp: Do
        ifail = 0

        Read (nin,*,Iostat=ifail)
        If (ifail==0) Then
          Read (nin,*,Iostat=ifail) nu
          If (ifail/=0) Then
            Exit d_lp
          End If
        Else
          Exit d_lp
        End If
        Read (nin,*) tail(1:n)
        Read (nin,*) a(1:n)
        Read (nin,*) b(1:n)
        Read (nin,*) delta(1:n)
        Read (nin,*)(rc(i,1:n),i=1,n)

!       Calculate probablity
        ifail = 0
        prob = g01hdf(n,tail,a,b,nu,delta,iscov,rc,ldrc,epsabs,epsrel,numsub, &
          nsampl,fmax,errest,ifail)

        Write (nout,99999) 'Probability:   ', prob
        Write (nout,99998) 'Error estimate:', errest
        Write (nout,*)
```

```
      End Do d_lp

99999 Format (2X,A24,E24.8)
99998 Format (2X,A24,E24.2)
    End Program g01hdfe
```

9.2 Program Data

```
G01HDF Example Program Data
5 1 : n iscov
Example 1
10.0  : nu
U U U U U : tails
-1.0e-1 -1.0e-1 -1.0e-1 -1.0e-1 -1.0e-1 : lower bounds
 888  888  888  888  888 : upper bounds
 0.0  0.0  0.0  0.0  0.0 : delta
1.0   0.75  0.75  0.75  0.75
0.75  1.0   0.75  0.75  0.75
0.75  0.75  1.0   0.75  0.75
0.75  0.75  0.75  1.0   0.75
0.75  0.75  0.75  0.75  1.0  : correlation matrix
Example 2
3.0  : nu
L L L L L : tails
 888  888  888  888  888 : lower bounds
-1.0e-1 -1.0e-1 -1.0e-1 -1.0e-1 -1.0e-1 : upper bounds
 1.0  2.0  3.0  3.0  3.0 : delta
1.0   0.75  0.75  0.75  0.75
0.75  1.0   0.75  0.75  0.75
0.75  0.75  1.0   0.75  0.75
0.75  0.75  0.75  1.0   0.75
0.75  0.75  0.75  0.75  1.0  : correlation matrix
```

9.3 Program Results

```
 G01HDF Example Program Results

           Probability:             0.30164222E+00
           Error estimate:                0.11E-04

           Probability:             0.86224881E-04
           Error estimate:                0.73E-07
```